Using BY Variable Text Substitution to Add Data Values to SAS/GRAPH® Titles and Footnotes

Earl Westerlund, Eastman Kodak Company, Rochester, NY

It is often desirable to add data values - regression coefficients, for example - to a SAS/GRAPH® title or footnote. This is easily accomplished with macro variables when the entire data set is being processed. When the data are to be grouped using a BY variable, the task becomes more difficult.

Beginning with release 6.07 of the SAS® System, the BY variable text substitution feature can be used to accomplish this task. The trick is to tell the system that the data set is sorted by the values that you wish to appear on the plot, in addition to the actual BY variables. This makes the variable names and values available to SAS/GRAPH.

Two cautions:
• The only way this technique will work correctly is if the values you wish to display do not change within a BY group. If they do, you will get a plot for each unique combination of values - or, the procedure will stop with a "not sorted" error message.
• You will probably want to turn off the default BY line and display one yourself. This is because the default BY line will contain all the display values, as well as the "real" BY variable value.

An example of this technique follows. It does a simple scatter plot, but calculates and displays the means of the X and Y variables in the footnote.

SAMPLE INPUT

```sas
/*** Read data ***/
DATA RAWDATA;
  INPUT ID $ X Y;
CARDS;
A  0.37 -1.21
A  0.59  0.95
B  -1.18 -3.54
B   5.24 -1.87
RUN;

/*** Compute means & std's ***/
PROC SUMMARY NWAY DATA=RAWDATA;
  CLASS ID;
  VAR X Y;
  OUTPUT OUT=STATS MEAN=XAVG YAVG STD=XSTD YSTD;
RUN;

/*** Add parameters to data set - define extra "sorted by" variables ***/
DATA ADDPARMS(SORTEDBY=ID XAVG YAVG XSTD YSTD);
  MERGE RAWDATA STATS;
  BY ID;
RUN;

/*** Turn off the BY line, then produce the plot! ***/
OPTIONS NOBYLINE;
PROC GPLOT DATA=ADDPARMS GOUT=KEEP;
  BY ID XAVG YAVG;
  TITLE F=SIMPLEX C=BLUE 'Example Using BY-Variable Text Substitution';
  TITLE2 H=1.2 '#BYVAR1=#BYVAL1';
  PLOT Y*X=1;
  SYMBOL I=NONE V=STAR C=BLUE;
  FORMAT XAVG YAVG 8.2;
  FOOTNOTE H=1.2 'X Mean=#BYVAL(XAVG); Y Mean=#BYVAL(YAVG)';
RUN;
```
Using BY-Variable Text Substitution

X Mean=-0.01; Y Mean=0.01

X Mean=2.58; Y Mean=2.00

REFERENCE


SAS/GRAPH and SAS are registered trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

AUTHOR

Earl Westerlund
Eastman Kodak Company
1669 Lake Avenue, Mail Code 24608
Rochester, NY 14652-4608
Phone (716) 722-1983
Internet EarlW@Kodak.COM

154